Problem:
Score : 600 points

Problem Statement

There are N cities in Republic of AtCoder. The size of the i-th city is A_{i}.
Takahashi would like to build N-1 bidirectional roads connecting two cities so that any city can be reached from any other city by using these roads.

Assume that the cost of building a road connecting the i-th city and the j-th city is |i-j| \times D + A_{i} + A_{j}.
For Takahashi, find the minimum possible total cost to achieve the objective.

Constraints

1 \leq N \leq 2 \times 10^5

1 \leq D \leq 10^9

1 \leq A_{i} \leq 10^9

A_{i} and D are integers.

Input

Input is given from Standard Input in the following format:

N D
A_1 A_2 ... A_N

Output

Print the minimum possible total cost.

Sample Input 1

3 1
1 100 1

Sample Output 1

106

This cost can be achieved by, for example, building roads connecting City 1, 2 and City 1, 3.

Sample Input 2

3 1000
1 100 1

Sample Output 2

2202

Sample Input 3

6 14
25 171 7 1 17 162

Sample Output 3

497

Sample Input 4

12 5
43 94 27 3 69 99 56 25 8 15 46 8

Sample Output 4

658

Solution in Common Lisp (SBCL 1.1.14) (judge verdict: Wrong Answer)

;; -*- coding: utf-8 -*-
(eval-when (:compile-toplevel :load-toplevel :execute)
  (sb-int:defconstant-eqx OPT
    #+swank '(optimize (speed 3) (safety 2))
    #-swank '(optimize (speed 3) (safety 0) (debug 0))
    #'equal)
  #+swank (ql:quickload '(:cl-debug-print :fiveam) :silent t)
  #-swank (set-dispatch-macro-character
           ;; enclose the form with VALUES to avoid being captured by LOOP macro
           #\# #\> (lambda (s c p) (declare (ignore c p)) `(values ,(read s nil nil t)))))
#+swank (cl-syntax:use-syntax cl-debug-print:debug-print-syntax)
#-swank (disable-debugger) ; for CS Academy

;; BEGIN_INSERTED_CONTENTS
;;;
;;; Disjoint set by Union-Find algorithm
;;;

(defstruct (disjoint-set
            (:constructor make-disjoint-set
                (size &aux (data (make-array size :element-type 'fixnum :initial-element -1))))
            (:conc-name ds-))
  (data nil :type (simple-array fixnum (*))))

(declaim (ftype (function * (values (mod #.array-total-size-limit) &optional)) ds-root))
(defun ds-root (disjoint-set x)
  "Returns the root of X."
  (declare (optimize (speed 3))
           ((mod #.array-total-size-limit) x))
  (let ((data (ds-data disjoint-set)))
    (if (< (aref data x) 0)
        x
        (setf (aref data x)
              (ds-root disjoint-set (aref data x))))))

(declaim (inline ds-unite!))
(defun ds-unite! (disjoint-set x1 x2)
  "Destructively unites X1 and X2 and returns true iff X1 and X2 become
connected for the first time."
  (let ((root1 (ds-root disjoint-set x1))
        (root2 (ds-root disjoint-set x2)))
    (unless (= root1 root2)
      (let ((data (ds-data disjoint-set)))
        ;; ensure the size of root1 >= the size of root2
        (when (> (aref data root1) (aref data root2))
          (rotatef root1 root2))
        (incf (aref data root1) (aref data root2))
        (setf (aref data root2) root1)))))

(declaim (inline ds-connected-p))
(defun ds-connected-p (disjoint-set x1 x2)
  "Returns true iff X1 and X2 have the same root."
  (= (ds-root disjoint-set x1) (ds-root disjoint-set x2)))

(declaim (inline ds-size))
(defun ds-size (disjoint-set x)
  "Returns the size of the connected component to which X belongs."
  (- (aref (ds-data disjoint-set)
           (ds-root disjoint-set x))))

(declaim (inline find-argopt))
(defun find-argopt (iterable predicate &key start end (key #'identity))
  "Returns an index (or key) x at which ITERABLE takes the minimal (or maximal,
depending on PREDICATE) value, and returns ITERABLE[x] as the second value.

To explain the behaviour briefly, when (FUNCALL PREDICATE (AREF ITERABLE
<index>) (AREF ITERABLE <index of current optimum>)) holds, <index> and <index
of current optimum> are swapped.

When ITERABLE is a hash-table, START and END are ignored."
  (declare ((or null (integer 0 #.most-positive-fixnum)) start end)
           ((or hash-table sequence) iterable))
  (labels ((invalid-range-error ()
             (error "Can't find optimal value in null interval [~A, ~A) on ~A" start end iterable)))
    (etypecase iterable
      (list
       (let* ((start (or start 0))
              (end (or end most-positive-fixnum))
              (iterable (nthcdr start iterable)))
         (when (or (null iterable)
                   (>= start end))
           (invalid-range-error))
         (let ((opt-element (car iterable))
               (opt-index start)
               (pos start))
           (dolist (x iterable)
             (when (>= pos end)
               (return-from find-argopt (values opt-index opt-element)))
             (when (funcall predicate
                            (funcall key x pos)
                            (funcall key opt-element pos))
               (setq opt-element x
                     opt-index pos))
             (incf pos))
           (values opt-index opt-element))))
      (vector
       (let ((start (or start 0))
             (end (or end (length iterable))))
         (when (or (>= start end)
                   (>= start (length iterable)))
           (invalid-range-error))
         (let ((opt-element (aref iterable start))
               (opt-index start))
           (loop for i from start below end
                 for x = (aref iterable i)
                 do (when (funcall predicate
                                   (funcall key x i)
                                   (funcall key opt-element i))
                      (setq opt-element x
                            opt-index i)))
           (values opt-index opt-element))))
      (hash-table
       (assert (and (null start) (null end)))
       (when (zerop (hash-table-count iterable))
         (invalid-range-error))
       (let* ((opt-value (gensym))
              (opt-key opt-value))
         (maphash
          (lambda (hash-key x)
            (when (or (eq opt-key opt-value)
                      (funcall predicate
                               (funcall key x hash-key)
                               (funcall key opt-value hash-key)))
              (setq opt-value x
                    opt-key hash-key)))
          iterable)
         (values opt-key opt-value))))))


(declaim (ftype (function * (values fixnum &optional)) read-fixnum))
(defun read-fixnum (&optional (in *standard-input*))
  "NOTE: cannot read -2^62"
  (macrolet ((%read-byte ()
               `(the (unsigned-byte 8)
                     #+swank (char-code (read-char in nil #\Nul))
                     #-swank (sb-impl::ansi-stream-read-byte in nil #.(char-code #\Nul) nil))))
    (let* ((minus nil)
           (result (loop (let ((byte (%read-byte)))
                           (cond ((<= 48 byte 57)
                                  (return (- byte 48)))
                                 ((zerop byte) ; #\Nul
                                  (error "Read EOF or #\Nul."))
                                 ((= byte #.(char-code #\-))
                                  (setf minus t)))))))
      (declare ((integer 0 #.most-positive-fixnum) result))
      (loop
        (let* ((byte (%read-byte)))
          (if (<= 48 byte 57)
              (setq result (+ (- byte 48)
                              (* 10 (the (integer 0 #.(floor most-positive-fixnum 10))
                                         result))))
              (return (if minus (- result) result))))))))

(defmacro dbg (&rest forms)
  #+swank
  (if (= (length forms) 1)
      `(format *error-output* "~A => ~A~%" ',(car forms) ,(car forms))
      `(format *error-output* "~A => ~A~%" ',forms `(,,@forms)))
  #-swank (declare (ignore forms)))

(defmacro define-int-types (&rest bits)
  `(progn
     ,@(mapcar (lambda (b) `(deftype ,(intern (format nil "UINT~A" b)) () '(unsigned-byte ,b))) bits)
     ,@(mapcar (lambda (b) `(deftype ,(intern (format nil "INT~A" b)) () '(signed-byte ,b))) bits)))
(define-int-types 2 4 7 8 15 16 31 32 62 63 64)

(declaim (inline println))
(defun println (obj &optional (stream *standard-output*))
  (let ((*read-default-float-format* 'double-float))
    (prog1 (princ obj stream) (terpri stream))))

(defconstant +mod+ 1000000007)

;;;
;;; Body
;;;

(defun main ()
  (declare (inline sort sb-impl::stable-sort-list))
  (let* ((n (read))
         (d (read))
         (as (make-array n :element-type 'uint31))
         edges)
    (declare (uint31 n d))
    (dotimes (i n)
      (setf (aref as i) (read-fixnum)))
    (labels ((calc-cost (i j)
               (declare (uint31 i j)
                        (values uint62 &optional))
               (+ (aref as i) (aref as j)
                  (* d (abs (- i j))))))
      (sb-int:named-let recur ((l 0) (r n))
        (declare (uint31 l r))
        (when (>= (- r l) 2)
          (let ((mid (ash (+ l r) -1)))
            (recur l mid)
            (recur mid r)
            (let ((pivot1
                    (find-argopt as #'< :start l :end mid :key (lambda (a i) (- a (* i d)))))
                  (pivot2
                    (find-argopt as #'< :start mid :end r :key (lambda (a i) (+ a (* i d))))))
              (loop for i from l below mid
                    do (push (cons i pivot2) edges))
              (loop for i from mid below r
                    do (push (cons i pivot1) edges))))))
      (setq edges (sort edges #'<
                        :key (lambda (edge)
                               (calc-cost (car edge) (cdr edge)))))
      (let ((dset (make-disjoint-set n))
            (res 0))
        (dolist (edge edges)
          (let ((i (car edge))
                (j (cdr edge)))
            (unless (ds-connected-p dset i j)
              (ds-unite! dset i j)
              (incf res (calc-cost i j)))))
        (println res)))))

#-swank (main)

;;;
;;; Test and benchmark
;;;

#+swank
(defun io-equal (in-string out-string &key (function #'main) (test #'equal))
  "Passes IN-STRING to *STANDARD-INPUT*, executes FUNCTION, and returns true if
the string output to *STANDARD-OUTPUT* is equal to OUT-STRING."
  (labels ((ensure-last-lf (s)
             (if (eql (uiop:last-char s) #\Linefeed)
                 s
                 (uiop:strcat s uiop:+lf+))))
    (funcall test
             (ensure-last-lf out-string)
             (with-output-to-string (out)
               (let ((*standard-output* out))
                 (with-input-from-string (*standard-input* (ensure-last-lf in-string))
                   (funcall function)))))))

#+swank
(defun get-clipbrd ()
  (with-output-to-string (out)
    ;; (run-program "C:/Windows/System32/WindowsPowerShell/v1.0/powershell.exe" '("get-clipboard") :output out)
    (run-program "powershell.exe" '("-Command" "Get-Clipboard") :output out :search t)))

#+swank (defparameter *this-pathname* (uiop:current-lisp-file-pathname))
#+swank (defparameter *dat-pathname* (uiop:merge-pathnames* "test.dat" *this-pathname*))

#+swank
(defun run (&optional thing (out *standard-output*))
  "THING := null | string | symbol | pathname

null: run #'MAIN using the text on clipboard as input.
string: run #'MAIN using the string as input.
symbol: alias of FIVEAM:RUN!.
pathname: run #'MAIN using the text file as input."
  (let ((*standard-output* out))
    (etypecase thing
      (null
       (with-input-from-string (*standard-input* (delete #\Return (get-clipbrd)))
         (main)))
      (string
       (with-input-from-string (*standard-input* (delete #\Return thing))
         (main)))
      (symbol (5am:run! thing))
      (pathname
       (with-open-file (*standard-input* thing)
         (main))))))

#+swank
(defun gen-dat ()
  (uiop:with-output-file (out *dat-pathname* :if-exists :supersede)
    (format out "")))

#+swank
(defun bench (&optional (out (make-broadcast-stream)))
  (time (run *dat-pathname* out)))

;; To run: (5am:run! :sample)
#+swank
(it.bese.fiveam:test :sample
  (it.bese.fiveam:is
   (common-lisp-user::io-equal "3 1
1 100 1
"
    "106
"))
  (it.bese.fiveam:is
   (common-lisp-user::io-equal "3 1000
1 100 1
"
    "2202
"))
  (it.bese.fiveam:is
   (common-lisp-user::io-equal "6 14
25 171 7 1 17 162
"
    "497
"))
  (it.bese.fiveam:is
   (common-lisp-user::io-equal "12 5
43 94 27 3 69 99 56 25 8 15 46 8
"
    "658
")))